



# AutoLISP

## Module 6

### User Input - Part 1

Why is it **Important** for you to learn this task?

#### RATIONALE:

Almost all AutoLISP programs require user input. This module will teach you how to request, use, display, and store user input with different data types.

Here is what you will be able to do when you complete each **Step** of this learning activity package:

#### OBJECTIVE(S):

1. Describe the string control characters used by AutoLISP.  
Describe and apply the function PRINC.  
Describe and apply the function GETINT.  
Describe and apply the function GETREAL.  
Describe and apply the function GETSTRING.

To show that you have **Mastered** this task, here is what you will be asked to do:

#### PERFORMANCE EVALUATION:

1. Complete Self-Test No. 1 with 100% accuracy.
2. Complete Lab Exercise L1150-01 with 100% accuracy.
3. Complete Lab Exercise L1150-02 with 100% accuracy.

# OBJECTIVE NO. 1

When you complete this objective, you will be able to:

- Describe the string control characters used by AutoLISP.
- Describe and apply the function PRINC.
- Describe and apply the function GETINT.
- Describe and apply the function GETREAL.
- Describe and apply the function GETSTRING.

Complete each of the learning activities listed below.

## LEARNING ACTIVITIES

**DO** the following things:

**USE** the following resources:

1. Read Information Sheet No. 1.
2. Complete Self-Test No. 1.
3. Check your answers to Self-Test No. 1 and correct any errors.
4. Complete Lab Exercise L1150-01.
5. Check your answers to Lab Exercise L1150-01 and correct any errors.
6. Complete Lab Exercise L1150-02.
7. Check your answers to Lab Exercise L1150-02 and correct any errors.

# INFORMATION SHEET NO. 1

## STRING CONTROL CHARACTERS

There are several string control characters that can be used in AutoLISP. A *string control character* is a special character, used inside of a string, that enables you to control how AutoLISP prints the string. All string control characters must follow the special character '\'. The string control characters that can be used in AutoLISP are shown in Table A-1150-1.

CODE	MEANING TO AUTOLISP
\n	A new Line (The same as pressing the ENTER key).
\\	The \ character.
\"	The " character.
\t	A tab.
\e	The escape character.
\r	The return character.
\nnn	Octal code with the number ' nnn '.
\u+nnnn	Unicode Character Sequence (Release 13)
\m+nnnnn	Multi-byte Character Sequence (Release 13)

**Table A-1150-1**  
String Control Characters

In this module, only the control characters '\n, \\, and '\" ' will be covered. The remainder will be covered in a later module.

## AutoLISP Function - PRINC

AutoCAD Release: All

### DESCRIPTION

The PRINC function is used to print an *expression* to the screen or to a file.

PRINC prints the control characters without expansion. This concept will be explained in much greater detail in a later module. For now, use PRINC to print messages or the values assigned to variables onto the screen.

The use of *file-description* will be covered in a later module.

### RETURNS

It returns the *expression*.

### FORMAT

(princ [*expression* [*file-description*]])

## EXAMPLES

### Example No. 1

Command: **(princ "ENTER THE DIAMETER: ")**  
ENTER THE DIAMETER: "ENTER THE DIAMETER: "

Command:

{In this example, the PRINC function prints the string ENTER THE DIAMETER onto the screen. It is printed without double quotations, but then returns the string "ENTER THE DIAMETER".}

### Example No. 2

Command: **(princ "\nENTER THE DIAMETER: ")**  
ENTER THE DIAMETER: "\nENTER THE DIAMETER: "

Command:

{In this example, the control character '\n' is added to the string. It tells AutoLISP that you want a new line entered before printing the expression. You can see that a new line is printed before the string is printed. You will find this option to be very valuable as you write your AutoLISP programs. Note how this functions returns the string "ENTER THE DIAMETER. In the next example you will see how this can be eliminated.}

### Example No. 3

Command: **(princ "\nENTER THE DIAMETER: ")(princ)**  
ENTER THE DIAMETER:

Command:

{In this example, the message ENTER THE DIAMETER is printed without any return. As you can see another (princ) was added as the last function. A (princ), without expression, returns nothing, therefore, it is used as the last statement in almost all AutoLISP programs to prevent AutoLISP from returning unwanted text to the screen.}

### Example No. 4

```
(defun testprinc1 ()  
  (setq num1 5  
        num2 3.75  
        str1 "Testing")  
  )  
  (princ num1)  
  (princ "\n")(princ num2)  
  (princ "\n")(princ str1)  
  (princ "\nThe End")  
  (princ)  
)
```

{When executed, this program looks as follows.}

Command: **(testprinc1)**

```
5  
3.75  
Testing  
The End
```

Command:

{As you can see, the PRINC function is used to print strings or the value assigned to variables. Note how the '\n' is used to print each new expression onto a new line.}

### Example No. 5

```
(defun testprinc2 (  
  (setq num1 25  
        num2 92.85  
        str1 "The value for num1 is: "  
  )  
  (princ str1)(princ num1)  
  (princ "\n\nThe value for \"num2\" is: ")(princ num2)  
  (princ "\n\nThe sum of num1 and num2 is: " ) (princ (+ num1 num2))  
  (princ "\n\nThe default directory is C:\\ACAD\\DRAW")  
  (princ)  
)
```

{When executed, this AutoLISP program will display the following.}

Command: **(testprinc2)**

```
The value for num1 is: 25  
The value for "num2" is: 92.85  
The sum of num1 and num2 is: 117.85  
The default directory is C:\\ACAD\\DRAW
```

Command:

{This example shows the use of PRINC and string control characters. Study the program and what it displays onto the screen.}

## AutoLISP Function T

AutoCAD Release: All

### DESCRIPTION

The 'T' function is used in an AutoLISP function to return True.

### EXAMPLES

#### Example No. 1

Command: **(princ T)**

```
TT
```

Command:

{This example shows how the function 'T' returns a True. You will see how to use this later in the module. There are two T's since it prints a 'T' and returns a 'T'.}

# AutoLISP Function - GETINT

AutoCAD Release: All

## DESCRIPTION

The GETINT function is used to obtain an integer from the user.

This function will pause and wait for an integer to be entered. The integer must be between '-32,768' and '32,767'. GETINT will only accept an integer. If the user enters any other data type in response to a GETINT prompt, AutoLISP issues an error message and reprompts the user.

## RETURNS

An integer.

## FORMAT

(getint [*prompt*])

## EXAMPLES

### Example No. 1

Command: (setq num1 (getint))

8

8

Command:

{ This example shows the GETINT function used without a prompt. Used this way, the program would stop and wait for the user to enter an integer. But because there is no prompt, the user would not know what to enter. This is a poor way to use this function. }

### Example No. 2

Command: (setq num2 (getint "\nEnter a number: "))

Enter a number: 34.3

Requires an integer value.

Enter a number: w

Requires an integer value.

Enter a number: 4

4

Command:

{ This example shows how AutoLISP will continue prompting the user until an integer is entered. Note how the integer is returned. }

# AutoLISP Function - GETREAL

AutoCAD Release: All

## DESCRIPTION

The GETREAL function is used to obtain a real number from the user.

This function will pause and wait for a real number to be entered. GETREAL will not accept any other data type except a real number. If the user enters any other data type, except an integer, in response to a GETREAL prompt, AutoLISP issues an error message and reprompts the user. If the user enters an integer, GETREAL will convert it into a real.

## RETURNS

A real number.

## FORMAT

(getreal [*prompt*])

## EXAMPLES

### Example No. 1

Command: (setq num2 (getreal))

2.5

2.5

Command:

{This example shows the GETREAL function used without a prompt. Used this way, AutoLISP would stop and wait for the user to enter a real number. But because there is no prompt, the user would not know what to enter. This is a poor way to use this function.}

### Example No. 2

Command: (setq num2 (getreal "\nEnter a real number: "))

Enter a real number: **D**

Requires a numeric value.

Enter a real: **4**

4.0

{This example show how GETREAL prompts the user and waits for a number. If the answer is an integer, it converts it into a real. If the response is a non-number, it issues an error message and reprompts the user.}

# AutoLISP Function - GETSTRING

AutoCAD Release: All

## DESCRIPTION

The GETSTRING function is used to obtain a string from the user.

This function will pause and wait for the user to enter characters from the keyboard. If the *cr* is present, and is not nil, the string can be entered with spaces and must be terminated with the ENTER key. If *cr* is not supplied, or is nil, the string will be terminated by either a space or an ENTER key.

If the string is longer than 132 characters, GETSTRING will return only the first 132 characters. If the user enters a '\' AutoLISP will automatically convert it into a double '\\'. This is done so that the correct string is used for a DOS path.

## RETURNS

A string.

## FORMAT

(getstring [*cr*] [*prompt*])

## EXAMPLES

### Example No. 1

Command: (setq str1 (getstring))

Hello

"Hello"

Command:

{This example shows the GETSTRING function used without a prompt. Used this way, AutoLISP would stop and wait for the user to enter a string. Since *cr* is not supplied, the user could not enter a space as part of the string. Both a space and an ENTER key would terminate the string.}

### Example No. 2

Command: (setq str1 (getstring "\nEnter Your SIR name: "))

Enter your SIR name: SMITH"SMITH"

Command:

{This example shows the GETSTRING function used without a *cr*. A space or an ENTER key will terminate the string. Note how it returns the string immediately after the user entry. This is because the user terminated it with a space rather than an ENTER.}

### Example No. 3

Command: (setq str2 (getstring T "\nEnter your name:"))

Enter your name: JOHN SMITH

"JOHN SMITH"

Command:

{In this example the *cr* is present and evaluates true. A 'T' can be used anywhere in an AutoLISP functions that requires a true value. Note how this function returns the string on a new line. This is because the user terminated the input with an ENTER key (means new line to AutoLISP).}

### Example No. 4

```
; ***** AutoCAD Modules *****
; A1150-01.LSP (v12)          Written by: J. Smith          040214
; *****
;
; This program contains two functions that will
; 1. Convert meters to feet
; 2. Convert feet to meters.
; This function will work like a command and prompt the user.
;
; ***** Function - M2F *****
; converts meters to feet
;
(defun C:M2F (/ meter1 feet1)
  (setq meter1 (getreal "\nEnter Meters "))
  (setq feet1 (* 3.280839 meter1))
  (princ "\n")(princ meter1)(princ " meters equals ")
  (princ feet1)(princ " feet")
  (princ)
)
;
; ***** Function - C:F2M *****
; converts feet to meters
;
(defun C:F2M (/ feet2 inch2 mets2)
  (setq feet2 (getint "\nEnter feet: "))
  (setq inch2 (getreal "\nEnter inches in decimals: "))
  (setq mets2 (* 0.3048 (+ feet2 (/ inch2 12.0))))
  (princ "\nThe number of meters is ")(princ mets2)
  (princ)
)
;
; ***** The End *****
```

{This program converts feet to meters and meters to feet. It shows how the PRINC function is used as well as the use of GETREAL and GETINT.}

# SELF-TEST NO. 1

## DIRECTIONS

1. Answer the following questions.
2. Compare your answers to the enclosed answer key.
3. If you disagree with any of the answers, review the learning material and/or check with your instructor.
4. If no problems arise, continue with the next step.

1. The GETSTRING function will pause and wait for the user to enter characters from the keyboard. If the *cr* is present, and is not nil, the string can be entered with \_\_\_\_\_ and must be terminated with the \_\_\_\_\_ key. If *cr* is not supplied, or is nil, the string will be terminated by either a \_\_\_\_\_ or an \_\_\_\_\_ key.

2. Beside each of the string control characters listed below, give a brief description of what effect each one has in a print function.

\n \_\_\_\_\_

\\ \_\_\_\_\_

\" \_\_\_\_\_

\t \_\_\_\_\_

\e \_\_\_\_\_

\r \_\_\_\_\_

\nnn \_\_\_\_\_

3. The GETINT function will pause and wait for an integer to be entered. Only an integer between the numbers \_\_\_\_\_ and \_\_\_\_\_ will be accepted.

4. All string control characters must follow the special character \_\_\_\_\_ .

5. If the user enters any other data type, except an integer which it converts to a real, in response to a GETREAL prompt, AutoLISP \_\_\_\_\_.

6. If the string is longer than \_\_\_\_\_ characters in response to the GETSTRING function, it will return only \_\_\_\_\_.

7. Given the following AutoLISP expression.

```
(princ "\nENTER THE DIAMETER: ")(princ)
```

What is the purpose the single (princ) function?

---

---

# LAB EXERCISE NO. 1

## Lab Exercise L1150-01

### Description

Program Name: L1150-01.LSP

1. Function 1 Name: DTORAD

Purpose: To convert degrees-minutes-seconds to decimal degrees and radians.

- a) Prompt the user to enter their name.
- b) Prompt the user to enter degrees as an integer only, the minutes as an integer only, and the seconds as a real number.
- c) After calculating the decimal degrees and radians, print the results on the screen.

2. Function 2 Name: dtor

Purpose: To convert degrees to radians.

- a) Do not rewrite this function. Copy the function you wrote in lab L1140-02.LSP into this program. Send your answer, as an argument, to this function to convert degrees to radians.

### Program Example

Command: **DTORAD**

Enter your FIRST name, a space, and then your LAST name: **JOHN SMITH**

Enter degrees: **67**

Enter minutes: **32**

Enter seconds.decimal: **39.45**

{The following lines are then printed to the screen.}

Your name is JOHN SMITH.

The conversion to decimal degrees is 67.5443

The number of radians is 1.17887

Command:

# LAB EXERCISE NO. 2

## Lab Exercise L1150-02

Description

Program Name: L1150-02.LSP

1. Function 1 Name: C:CIRAREA

Purpose: To calculate the area of a circle.

- a) Formula:  $\text{Area} = \text{PI} \times \text{radius}^2$
- b) Prompt the user to enter the diameter of the circle. Allow a decimal value to be entered.

### Program Example

Command: **CIRAREA**  
Enter diameter of circle: **34.5**  
The area of the circle is 934.82

Command:

2. Function 2 Name: C:CIRPERM

Purpose: To calculate the circumference of a circle

- a) Formula:  $\text{Circumference} = \text{PI} \times \text{diameter}$
- b) Prompt the user to enter the diameter of the circle. Allow a decimal value to be entered.

### Program Example

Command: CIRPERM  
Enter diameter of circle: **5.67**  
The circumference of the circle is 17.8128

Command:

# ANSWER KEY

## SELF-TEST # 1

1. The GETSTRING function will pause and wait for the user to enter characters from the keyboard. If the *cr* is present, and is not nil, the string can be entered with **spaces** and must be terminated with the **ENTER** key. If *cr* is not supplied, or is nil, the string will be terminated by either a **space** or an **ENTER** key.
2. Beside each of the string control characters listed below, give a brief description of what effect each one has in a print function.

`\n`            **A new Line or the same as pressing the Enter key.**

`\\`            **The \ character**

`\"`            **The " character**

`\t`            **A tab**

`\e`            **The escape character**

`\r`            **The return character**

`\nnn`        **Octal code nnn**

3. The GETINT function will pause and wait for an integer to be entered. Only an integer between the numbers **-32,768** and **32,767** will be accepted.
4. All string control characters must follow the special character '`\`'.
5. If the user enters any other data type, except an integer which it converts to a real, in response to a GETREAL prompt, AutoLISP **issues an error message and reprompts the user.**
6. If the string is longer than **132** characters in response to the GETSTRING function, it will return only **the first 132 characters.**
7. Given the following AutoLISP expression.

```
(princ "\nENTER THE DIAMETER: ")(princ)
```

What is the purpose the single (princ) function?

**It forces AutoLISP to returns nothing, this stops the first PRINC function from displaying its return onto the screen.**

## LAB EXERCISE No. 1 - L1150-01

```
***** AutoCAD Modules *****
; L1150-01.LSP (v12)           Written by: J. Smith           040215
*****
; This program convert degrees/minutes/seconds to radians
; ***** Function - DTORAD *****
(defun C:DTORAD (/ deg1 min1 sec1 degd rad1)
  (setq name1 (getstring T "\nEnter your FIRST name, a space, and then your LAST name: "))
  (setq deg1 (getint "\nEnter degrees: "))
  (setq min1 (getint "\nEnter minutes: "))
  (setq sec1 (getreal "\nEnter seconds.decimal: "))
  (setq degd (+ deg1 (/ min1 60.0) (/ sec1 3600.0)))
  (setq rad1 (dtor degd))
  (princ "\nYour name is ")(princ name1)(princ ".")
  (princ "\nThe conversion to decimal degrees is ")
  (princ degd)(princ "\nThe number of radians is ")(princ rad1)
  (princ)
)
; ***** Function - dtor *****
; converts degrees (decimals) to radians
;
(defun dtor (angd)
  (* pi (/ angd 180.0))
)
; ***** The End *****
```

## LAB EXERCISE No. 2 - L1150-02

```
***** AutoCAD Modules *****
; L1150-02.LSP (v12)           Written by: J. Smith           040214
*****
; This program contains two functions that will
; 1. Calculate the area of a circle and
; 2. Calculate the circumference of a circle.
; ***** Function - C:CIRAREA *****
; Calculates the area of a circle
;
(defun C:CIRAREA (/ diam1 cira1)
  (setq diam1 (getreal "\nEnter diameter of circle: "))
  (setq cira1 (* pi (* (/ diam1 2) (/ diam1 2))))
  (princ "\nThe area of the circle is ")(princ cira1)
  (princ)
)
; ***** Function C:CIRPERM *****
; calculates the circumference of a circle
;
(defun C:CIRPERM (/ diam1 circ1)
  (setq diam1 (getreal "\nEnter diameter of circle: "))
  (setq circ1 (* pi diam1))
  (princ "\nThe circumference of the circle is ")(princ circ1)
  (princ)
)
; ***** The End *****
```